

NFA-ACT-95-09

Program MOX
A Tool for the Calculation of
Nuclide Densities in MOX Fuels

J.L. Kloosterman

NFA memos are working documents that can be extended, changed or withdrawn without notice. These documents are internal papers, not meant for further distribution or citation.

This memo was made under contract number MEFIS 59436 (ECN projectnumber 1.1742) with the Dutch Ministry of Economic Affairs.

Abstract

This note describes the input, the output and the calculational method of the program MOX, which can be used to calculate the atomic densities in MOX fuels.

Distribution

E.E. Bende	1
J.H. Bultman	2
M.C. Duijvestijn	3
W.E. Freudenreich	4
J.A. Hendriks	5
A. Hogenbirk	6
H.Th. Klippel	7
A.J. Koning	8
J.C. Kuijper	9
J. Oppe	10
T.T.J.M. Peeters	11
B.J. Pijlgroms	12
C. Sciolla	13
R.C.L. van der Stad	14
E.C. Verkerk	15
Author	16 – 20

CONTENTS

1. INTRODUCTION	5
2. INPUT DESCRIPTION	6
3. OUTPUT	7
4. CALCULATIONAL METHOD	9
REFERENCES	11
APPENDIX A. FORTRAN SOURCE CODE	13

1. INTRODUCTION

The use of Mixed OXide (MOX) fuel in Light Water-cooled Reactors (LWRs) or Liquid Metal-cooled Reactors (LMRs) is a suitable way to recycle plutonium. However, the calculation of the isotopic composition of MOX fuel is not straightforward. Therefore, a program called MOX has been made to calculate the isotopic composition of both enriched UO_2 fuel and MOX fuel.

Chapter 2 of this note describes the input needed by the program. The output is described in chapter 3 and the calculational method in chapter 4.

2. INPUT DESCRIPTION

Program MOX needs 11 input data as given in table 2.1.

Table 2.1 *Values of the input parameters of the program MOX.*

Parameter	Explanation
RHO	The density of the fuel including the oxygen. Its units are g cm^{-3} .
WU4	The enrichment of the uranium in ^{234}U . This is the <i>weight</i> fraction of the ^{234}U in the uranium.
WU5	The enrichment of the uranium in ^{235}U . This is the <i>weight</i> fraction of the ^{235}U in the uranium.
WU8	The enrichment of the uranium in ^{238}U . This is the <i>weight</i> fraction of the ^{238}U in the uranium.
The program renormalizes the sum of WU4, WU5 and WU8 to one.	
WPU	The enrichment of the fuel in plutonium. This is the <i>weight</i> fraction of the plutonium in the sum of plutonium and uranium used.
FPU8	The <i>isotopic</i> fraction of ^{238}Pu in the plutonium.
FPU9	The <i>isotopic</i> fraction of ^{239}Pu in the plutonium.
FPU0	The <i>isotopic</i> fraction of ^{240}Pu in the plutonium.
FPU1	The <i>isotopic</i> fraction of ^{241}Pu in the plutonium.
FPU2	The <i>isotopic</i> fraction of ^{242}Pu in the plutonium.
FAM1	The <i>isotopic</i> fraction of ^{241}Am in the plutonium. Usually produced by decay of ^{241}Pu .
The program renormalizes the sum of FPU8 till FAM1 to one.	

When MOX is started it reads the default values for all input parameters from the file MOX.sys. If this file does not exist, it is created the first time MOX is run. When the file already exists, it is updated with the input data from the latest run.

The program can also be used by typing the values for all input parameters on the input command line, which facilitates the use of MOX in a shell script. These values are also used to update the data in the file MOX.sys.

3. OUTPUT

The output of the program is self-explanatory. A typical example of the output with input values given in table 3.1 is given hereafter. Note that the fractions given in table 3.1 sum to one. However, as explained in the previous section, this is not obligatory and the same output can be obtained when all fractions of the uranium or of the plutonium isotopes are scaled with a fixed number.

Table 3.1 *Example input for the program MOX.*

Parameter	Value
RHO	10.02
WU4	0.0
WU5	0.25
WU8	99.75
WPU	10.15
FPU8	4.03
FPU9	50.55
FPU0	22.97
FPU1	13.43
FPU2	9.02
FAM1	0.0

The output of the program when run in interactive mode is:

```

PROGRAM MOX
Calculation of nuclide densities for MOX fuels
Author J.L. Kloosterman
Version 1.00
Date 95/12/20

nuclide atom_mass atom_frac wght_frac density atom_dens
          (%)          (%)          (g/cm3) (at/barn.cm)
922340  234.0409   0.0000   0.0000   0.0000  0.00000E+00
922350  235.0439   0.2532   0.2500   0.0198  5.08382E-05
922380  238.0508   99.7468  99.7500   7.9170  2.00282E-02
920000  238.0432  100.0000 100.0000   7.9369  2.00790E-02
942380  238.0496    4.0300   4.0009   0.0359  9.07475E-05
942390  239.0522   50.5500  50.3961   0.4518  1.13828E-03
942400  240.0538   22.9700  22.9960   0.2062  5.17238E-04
942410  241.0568   13.4300  13.5014   0.1211  3.02417E-04
942420  242.0587    9.0200   9.1056   0.0816  2.03112E-04
952410  241.0568    0.0000   0.0000   0.0000  0.00000E+00
940000  239.7822  100.0000 100.0000   0.8966  2.25180E-03
 80160  15.9994  100.0000 100.0000   1.1866  4.46617E-02
total                                     10.0200

```

If the input parameters are given on the command line, the output is:

922340	0.00000E+00
922350	5.08382E-05
922380	2.00282E-02
942380	9.07475E-05
942390	1.13828E-03
942400	5.17238E-04
942410	3.02417E-04
942420	2.03112E-04
952410	0.00000E+00
80160	4.46617E-02

4. CALCULATIONAL METHOD

After reading all the input data, the effective atomic mass of the uranium (AU) is calculated according to:

$$AU = \frac{WU4 + WU5 + WU8}{\frac{WU4}{AU4} + \frac{WU5}{AU5} + \frac{WU8}{AU8}} \quad (4.1)$$

where $AU4$, $AU5$ and $AU8$ are the atomic masses of the uranium isotopes ^{234}U , ^{235}U and ^{238}U , respectively, for which the values are given in table 4.1. The effective atomic mass of the plutonium (APU) is calculated according to:

$$APU = FPU8 \cdot APU8 + FPU9 \cdot APU9 + FPU0 \cdot APU0 + \\ FPU1 \cdot APU1 + FPU2 \cdot APU2 + FAM1 \cdot AAM1 \quad (4.2)$$

where $APU8$ till $APU2$ are the atomic masses of the plutonium isotopes ^{238}Pu till ^{242}Pu , respectively, and $AAM1$ is the atomic mass of ^{241}Am . Values for these parameters are given in table 4.1. The atomic fractions of the uranium isotopes are calculated according to:

$$FU4 = \frac{\frac{WU4}{AU4}}{\frac{WU4}{AU4} + \frac{WU5}{AU5} + \frac{WU8}{AU8}} \quad (4.3)$$

$$FU5 = \frac{\frac{WU5}{AU5}}{\frac{WU4}{AU4} + \frac{WU5}{AU5} + \frac{WU8}{AU8}} \quad (4.4)$$

$$FU8 = \frac{\frac{WU8}{AU8}}{\frac{WU4}{AU4} + \frac{WU5}{AU5} + \frac{WU8}{AU8}} \quad (4.5)$$

The atomic density of the uranium (NU) is calculated according to:

$$NU = \frac{N_{Avo} \cdot \rho_{fuel}}{\frac{AU}{1-WPU} + 2 \cdot AO \left(1 + \frac{WPU}{1-WPU} \cdot \frac{AU}{APU}\right)} \quad (4.6)$$

where N_{Avo} is Avogadro's number ($0.60221367 \cdot 10^{24}$ [1]) and ρ_{fuel} is the specific density of the fuel. The parameter AO is the atomic mass of the oxygen for which the value is given in table 4.1. The atomic density of the plutonium (NPU) is calculated according to:

$$NPU = \frac{WPU}{1-WPU} \cdot \frac{AU}{APU} \cdot NU \quad (4.7)$$

and that of the oxygen (NO) according to:

$$NO = 2 \cdot (NU + NPU) \quad (4.8)$$

With above-mentioned atomic densities and the atomic fractions $FU4$, $FU5$, $FU8$, $FPU8$, $FPU9$, $FPU0$, $FPU1$, $FPU2$ and $FAM1$, the atomic densities of each actinide isotope can easily be calculated.

The atomic masses of nuclides used in the program are programmed in data statements and given in table 4.1.

Table 4.1 *Values of atomic masses used in the program MOX [1].*

Parameter	Atomic mass
AO	15.9994
AU4	234.040946
AU5	235.043924
AU8	238.050784
APU8	238.049554
APU9	239.052157
APU0	240.053808
APU1	241.056845
APU2	242.058737
AAM1	241.056823

REFERENCES

- [1] David R. Lide. Handbook of Chemistry and Physics. CRC Press, 1993-1994.

APPENDIX A. FORTRAN SOURCE CODE

```
      program MOX
c=====
c   calculates atomic densities of U and Pu isotopes in MOX fuels
c   made by J.L. Kloosterman, ECN, Petten, The Netherlands
c=====
      parameter (maxu=3,maxpu=6,maxarg=11)
      character carg*80,csys*7
      logical logic
      integer iargc
      dimension au(maxu),fu(maxu),wu(maxu),iu(maxu),
1 apu(maxpu),fpu(maxpu),ipu(maxpu),carg(maxarg)
c-----next values from Handbook of Physics and Chemistry 74th edition
c-----be sure that the last uranium isotope is U-238
      data avo/0.60221367/,aoxyg/15.9994/,
1 au/234.040946,235.043924,238.050784/,
2 apu/238.049554,239.052157,240.053808,241.056845,242.058737,
3 241.056823/,
4 iu/922340,922350,922380/,io/80160/,
5 ipu/942380,942390,942400,942410,942420,952410/
      data csys/'MOX.sys',lsys/10/
c=====
c   read input data from file csys
c=====
      inquire(file=csys,exist=logic)
      if (logic) then
         open(unit=lsys,file=csys,form='unformatted',status='old')
         read(lsys) dens,(wu(i),i=1,maxu),wpu,(fpu(i),i=1,maxpu)
      else
         open(unit=lsys,file=csys,form='unformatted',status='new')
      endif
c=====
c   read input data from input command line if present
c=====
      narg = iargc()
      if (narg.gt.0 .and. narg.ne.maxarg) then
         write(6,1000) narg,maxarg
         close(lsys)
         stop
      endif
      do 10, i=1,narg
         call getarg(i,carg(i))
10  continue
c=====
c   read data
c=====
c-----read density of fuel (g/cm3)
      if (narg .eq. 0) then
         write(6,1100) dens
         call rdefr(5,dens)
      else
         read(carg(1),*) dens
      endif
c-----read weight frac (%) of uranium isotopes in uranium metal
      sum = 0.
      do 20, i=1,maxu
         j = i+1
         if (narg .eq. 0) then
            write(6,1110) iu(i),wu(i)
            call rdefr(5,wu(i))
         else
```

```

        read(carg(j),*) wu(i)
    endif
    sum = sum+wu(i)
20  continue
    do 25, i=1,maxu
        wu(i) = wu(i)/sum
25  continue
c-----read weight frac (%) of plutonium in total heavy metal (u+pu)
    if (narg .eq. 0) then
        write(6,1120) wpu
        call rdefr(5,wpu)
    else
        read(carg(maxu+2),*) wpu
    endif
    wpu = wpu/100.
c-----read atomic frac (%) of plutonium isotopes in total plutonium
    sum = 0.
    do 30, i=1,maxpu
        j = i+maxu+2
        if (narg .eq. 0) then
            write(6,1130) ipu(i),fpu(i)
            call rdefr(5,fpu(i))
        else
            read(carg(j),*) fpu(i)
        endif
        sum = sum+fpu(i)
30  continue
    do 35, i=1,maxpu
        fpu(i) = fpu(i)/sum
35  continue
c=====
c  calculate atomic fractions of uranium isotopes in uranium (fu)
c  calculate atomic mass of mix of uranium isotopes (auran)
c=====
    temp = 0.
    do 40, i=1,maxu
        temp = temp + wu(i)/au(i)
40  continue
    auran = 0.
    do 50, i=1,maxu
        fu(i) = wu(i)/(au(i)*temp)
        auran = auran + wu(i)/temp
50  continue
c=====
c  calculate atomic mass of mix of plutonium isotopes (APU)
c=====
    aplut = 0.
    do 60, i=1,maxpu
        aplut = aplut + fpu(i)*apu(i)
60  continue
c=====
c  calculate atomic density of uranium (curan)
c  calculate atomic density of plutonium (cplut)
c  calculate atomic density of oxygen (coxyg)
c=====
    x      = 1.-wpu
    y      = wpu/x
    z      = auran/x + 2*aoxyg*(1.+y*auran/aplut)
    curan  = avo*dens/z
    cplut  = y*curan*auran/aplut
    coxyg  = 2.*(curan+cplut)
c=====
c  calculate and print atomic densities uranium

```

```

c    calculate and print atomic densities plutonium
c    calculate and print atomic densities oxygen
c=====
      if (narg .eq. 0) write(6,1140)
      sum1 = 0.
      sum2 = 0.
      sum3 = 0.
      sum4 = 0.
      do 70, i=1,maxu
        if (narg .eq. 0) then
          ad = fu(i)*curan
          wd = ad*au(i)/avo
          write(6,1160) iu(i),au(i),100.*fu(i),100.*wu(i),wd,ad
          sum1 = sum1 + 100.*fu(i)
          sum2 = sum2 + 100.*wu(i)
          sum3 = sum3 + wd
          sum4 = sum4 + ad
        else
          write(6,1150) iu(i),fu(i)*curan
        endif
70    continue
      total = sum3
      if (narg.eq.0) write(6,1160) 920000,auran,sum1,sum2,sum3,sum4
      sum1 = 0.
      sum2 = 0.
      sum3 = 0.
      sum4 = 0.
      do 80, i=1,maxpu
        if (narg .eq. 0) then
          tp = 100.*fpu(i)*apu(i)/aplut
          ad = fpu(i)*cplut
          wd = ad*apu(i)/avo
          write(6,1160) ipu(i),apu(i),100.*fpu(i),tp,wd,ad
          sum1 = sum1 + 100.*fpu(i)
          sum2 = sum2 + tp
          sum3 = sum3 + wd
          sum4 = sum4 + ad
        else
          write(6,1150) ipu(i),fpu(i)*cplut
        endif
80    continue
      total = total + sum3
      if (narg .eq. 0) then
        wd = coxyg*aoxyg/avo
        write(6,1160) 940000,aplut,sum1,sum2,sum3,sum4
        write(6,1160) io,aoxyg,100.,100.,wd,coxyg
        write(6,1170) total+wd
      else
        write(6,1150) io,coxyg
      endif
c=====
c    write defaults in file csys
c=====
      rewind(lsys)
      write(lsys) dens, (100.*wu(i),i=1,maxu),100.*wpu,
1 (100.*fpu(i),i=1,maxpu)
      close(lsys)
c=====
1000 format(' error in MOX',/,
1 ' number of arguments',i4,' larger then maximum',i4)
1100 format(' give fuel density (default',f10.4,' g/cm3) ')
1110 format(' give weight perc of 'i6,' in uranium',
1 ' (default',f9.4,' %)')

```

```
1120 format(' give weight perc of plutonium in total heavy metal',
1 ' (default',f9.4,' %)' )
1130 format(' give atomic perc of 'i6,' in plutonium',
1 ' (default',f9.4,' %)' )
1140 format(' Program MOX',/,
1 ' Calculation of nuclide densities for MOX fuels',/,
2 ' Author J.L. Kloosterman',/,
3 ' Version 1.00 ',/,
4 ' Date 95/12/20 ',/,/,
1 ' nuclide atom_mass atom_frac wght_frac density atom_dens',/,
2 ' (%)(%)(g/cm3)(at/barn.cm)' )
1150 format(1x,i6,5x,1pe12.5)
1160 format(1x,i6,4(1x,f9.4),2x,1pe12.5)
1170 format(' total',31x,f9.4)
c=====
end
```